```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

Let us consider the model of a fluid in a tank satisfying Saint-Venant's equations and subjected to a one-dimensional horizontal move studied in F. Dubois, N. Petit, P. Rouchon, "Motion planning and nonlinear simulations for a tank containing a fluid", in the proceedings of the 5^{th} *European Control Conference*, Karlsruhe (Germany), 1999, and defined by the following system matrix:

```
> A:=DefineOreAlgebra(diff=[d,t],dual_shift=[delta,s],polynom=[s,t]):
> R:=matrix(2,3,[delta^2,1,-2*d*delta,1,delta^2,-2*d*delta]);
```

$$R := \begin{bmatrix} \delta^2 & 1 & -2\,d\,\delta \\ 1 & \delta^2 & -2\,d\,\delta \end{bmatrix}$$

Let us compute the endomorphism ring $E = \mathrm{end}_A(M)$ of the A-module $M = A^{1\times 3}/(A^{1\times 2}\,R)$, where $A = \mathbb{Q}[d,\delta]$ is the commutative polynomial ring of differential time-delay operators:

```
> Endo:=MorphismsConstCoeff(R,R,A,mult_table):
```

The A-module E is generated by the f_i's defined by $f_i(\pi(\lambda)) = \pi(\lambda\,P_i)$, where $\pi : A^{1\times 3} \longrightarrow M$ denotes the projection onto M, $\lambda \in A^{1\times 3}$ and the matrix $P_i \in A^{3\times 3}$ is one of the following matrices:

```
> Endo[1];
```

$$\left[\begin{bmatrix} 0 & 0 & 2\,d\,\delta \\ 0 & 0 & 2\,d\,\delta \\ 0 & 0 & \delta^2+1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 2\,d & -2\,d & 0 \\ \delta & -\delta & 0 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix} \right]$$

The generators $\{f_i\}_{i=1,\dots,4}$ of the A-module E satisfy the relations $Endo[2]\,F = 0$, with the notation $F = (f_1 \ \dots \ f_4)^T$, and $Endo[2]$ is the matrix defined by:

```
> Endo[2];
```

$$\begin{bmatrix} -1 & 1 & 0 & \delta^2 \\ -1 & \delta^2 & 0 & 1 \\ 0 & 0 & \delta^2-1 & 0 \end{bmatrix}$$

The multiplication table T of the generators $\{f_i\}_{i=1,\dots,4}$ is defined by $F \otimes F = T\,F$, where $\otimes$ denotes the Kronecker product, namely, $F \otimes F = ((f_1 \circ F)^T \ \dots \ (f_4 \circ F)^T)^T$, and T is the matrix $Endo[3]$ without the first column which corresponds to the indices (i,j) of the product $f_i \circ f_j$:

```
> Endo[3];
```

$$\begin{bmatrix} [1,1] & \delta^2+1 & 0 & 0 & 0 \\ [1,2] & 1 & 0 & 0 & 0 \\ [1,3] & 0 & 2\,d & 2 & -2\,d \\ [1,4] & 1 & 0 & 0 & 0 \\ [2,1] & 1 & 0 & 0 & 0 \\ [2,2] & 0 & 1 & 0 & 0 \\ [2,3] & 0 & 0 & 1 & 0 \\ [2,4] & 0 & 0 & 0 & 1 \\ [3,1] & 0 & 0 & 0 & 0 \\ [3,2] & 0 & 0 & 1 & 0 \\ [3,3] & 0 & 0 & -2\,d & 0 \\ [3,4] & 0 & 0 & -1 & 0 \\ [4,1] & 1 & 0 & 0 & 0 \\ [4,2] & 0 & 0 & 0 & 1 \\ [4,3] & 0 & 2\,d & 1 & -2\,d \\ [4,4] & 0 & 1 & 0 & 0 \end{bmatrix}$$

Let us compute idempotents of E, namely, elements $e \in E$ satisfying $e^2 = e$:

```
> Idem:=IdempotentsConstCoeff(R,Endo[1],A,0);
```

$$Idem := [[\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1/2 & 1/2 & 0 \\ 1/2 & 1/2 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1/2 & -1/2 & 0 \\ -1/2 & 1/2 & 0 \\ 0 & 0 & 0 \end{bmatrix}],$$

$$[Ore\_algebra, [\text{"}diff\text{"}, dual\_shift], [t, s], [d, \delta], [s, t], [], 0, [], [], [t, s], [], [],$$
$$[diff = [d, t], dual\_shift = [\delta, s]]]]$$

We obtain the two trivial idempotents 0 and id_M of E but also two other non-trivial idempotents e and f satisfying the relation $e + f = \mathrm{id}_M$. Let us consider the first non-trivial idempotent e of E defined by $e(\pi(\lambda)) = \pi(\lambda\, P)$, for all $\lambda \in A^{3\times 3}$, where P is the third matrix of $Idem[1]$ and $Q \in A^{2\times 2}$ is a matrix satisfying $R\,P = Q\,R$:

```
> P:=Idem[1,3]; Q:=Factorize(Mult(R,P,A),R,A);
```

$$P := \begin{bmatrix} 1/2 & 1/2 & 0 \\ 1/2 & 1/2 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad Q := \begin{bmatrix} 1/2 & 1/2 \\ 1/2 & 1/2 \end{bmatrix}$$

As the entries of the matrices P and Q belong to $\mathbb{Q}$, we can compute their Jordan normal forms:

```
> J[1]:=jordan(P,'W'); evalm(W);
```

$$J_1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad \begin{bmatrix} 1/2 & 1/2 & 0 \\ -1/2 & 1/2 & 0 \\ 0 & 1 & 1 \end{bmatrix}$$

```
> J[2]:=jordan(Q,'Z'); evalm(Z);
```

$$J_2 := \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix} \quad \begin{bmatrix} 1/2 & 1/2 \\ -1/2 & 1/2 \end{bmatrix}$$

Hence, we have $J_1 = W^{-1}\, P\, W$ and $J_2 = Z^{-1}\, Q\, Z$, and thus, the matrix R is equivalent to the block-matrix $Z^{-1}\, R\, W$ defined by:

```
> R_dec:=simplify(Mult(inverse(Z),R,W,A));
```

$$R\_dec := \begin{bmatrix} \delta^2 - 1 & 0 & 0 \\ 0 & 1 + \delta^2 - 4\, d\, \delta & -4\, d\, \delta \end{bmatrix}$$

We can simplify the previous matrix by post-multiplying it by the following unimodular matrix

```
> Y:=evalm(diag(1,evalm([[1,0],[-1,1]])));
```

$$Y := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & -1 & 1 \end{bmatrix}$$

in order to obtain the following simple block-diagonal matrix:

```
> R_final:=Mult(R_dec,Y,A);
```

$$R\_final := \begin{bmatrix} \delta^2 - 1 & 0 & 0 \\ 0 & \delta^2 + 1 & -4\, d\, \delta \end{bmatrix}$$

Hence, we obtain that the A-module can be decomposed as $M \cong M_1 \oplus M_2$, with the notations $M_1 = A/(A\,(\delta^2-1))$ and $M_2 = A^{1\times 2}/(A\,(\delta^2+1 \quad -4\,d\,\delta))$. Hence, if $\mathcal{F}$ denotes an A-module (e.g., $\mathcal{F} = C^{\infty}(\mathbb{R})$), then we have $\ker_{\mathcal{F}}(R.) \cong \ker_{\mathcal{F}}((\delta^2-1).) \oplus \ker_{\mathcal{F}}((\delta^2+1 \quad -4\,d\,\delta).)$. We note that $\ker_{\mathcal{F}}((\delta^2-1).)$ is formed by the 2-periodic functions of $\mathcal{F}$.

Let us study the A-module structure $A^{1\times 4}/(A^{1\times 3}\, Endo[2])$ of the endomorphism ring E:

```
> ext1:=Exti(Involution(Endo[2],A),A,1);
```

$$ext1 := \left[\begin{bmatrix} \delta^2 - 1 & 0 & 0 \\ 0 & \delta^2 - 1 & 0 \\ 0 & 0 & \delta^2 - 1 \end{bmatrix}, \begin{bmatrix} -1 & 0 & 0 & \delta^2 + 1 \\ 0 & 1 & 0 & -1 \\ 0 & 0 & 1 & 0 \end{bmatrix}, \begin{bmatrix} \delta^2 + 1 \\ 1 \\ 0 \\ 1 \end{bmatrix} \right]$$

Hence, we obtain that the following torsion elements of E

$$\begin{cases} t_1 = -f_1 + (\delta^2+1)\, f_4, \\ t_2 = f_2 - f_4, \\ t_3 = f_3, \end{cases} \qquad (\delta^2 - 1)\, t_i = 0, \quad i = 1, 2, 3,$$

generate the A-module $t(E)$. Moreover, we have $E/t(E) = A^{1\times 4}/(A^{1\times 3}\, ext1[2]) \cong A^{1\times 4}\, ext1[3]$, where $ext1[2]$ (resp., $ext1[3]$) denotes the second (resp., third) matrix of $ext1$. As the matrix $ext1[3]$ admits the following left-inverse over A

```
> LeftInverse(ext1[3],A);
```

$$\begin{bmatrix} 0 & 0 & 0 & 1 \end{bmatrix}$$

the A-module $E/t(E)$ is a free A-module of rank 1. The short exact sequence of A-modules

$$0 \longrightarrow t(E) \xrightarrow{\iota} E \xrightarrow{\rho} E/t(E) \longrightarrow 0,$$

ending with a projective A-module, splits, a fact implying:

$$E \cong t(E) \oplus E/t(E) \cong t(E) \oplus A.$$

Let us now study the A-module $t(E) = (A^{1\times 4}\, ext1[2])/(A^{1\times 2}\, Endo[2])$:

```
> L:=Factorize(Endo[2],ext1[2],A);
```

$$L := \begin{bmatrix} 1 & 1 & 0 \\ 1 & \delta^2 & 0 \\ 0 & 0 & \delta^2 - 1 \end{bmatrix}$$

```
> SyzygyModule(ext1[2],A);
```

$$INJ\,(3)$$

By Lemma 3.1 of T. Cluzeau, A. Quadrat, "Factoring and decomposing a class of linear functional systems", *Linear Algebra and Its Applications*, 428 (2008), 324-381, we obtain

$$t(E) \cong A^{1\times 3}/(A^{1\times 3}\, L) \cong A^{1\times 2}/(A^{1\times 2}\, Q) \oplus A/(A\,(\delta^2 - 1)),$$

where the matrix $Q \in A^{2\times 2}$ is defined by:

```
> Q:=submatrix(L,1..2,1..2);
```

$$Q := \begin{bmatrix} 1 & 1 \\ 1 & \delta^2 \end{bmatrix}$$

The matrix Q admits an equivalent diagonal matrix which can be computed as follows:

```
> Endo_Q:=MorphismsConstCoeff(Q,Q,A):
> Idem_Q:=IdempotentsMatConstCoeff(Q,Endo_Q[1],A,0,alpha);
```

$$Idem\_Q := [[\begin{bmatrix} 0 & -1 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}], [Ore\_algebra, [``diff", dual\_shift],$$
$$[t, s], [d, \delta], [s, t], [], 0, [], [], [t, s], [], [], [diff = [d, t], dual\_shift = [\delta, s]]]]$$

```
> F:=Idem_Q[1,1];
```

$$F := \begin{bmatrix} 0 & -1 \\ 0 & 1 \end{bmatrix}$$

```
> HeuristicDecomposition(Q,F,A)[1];
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 - \delta^2 \end{bmatrix}$$

Hence, we obtain that $A^{1\times 2}/(A^{1\times 2}\, Q) \cong A/(A\,1) \oplus A/(A\,(\delta^2 - 1)) \cong A/(A\,(\delta^2 - 1))$, a fact finally proving that the A-module E satisfies

$$E \cong [A/(A\,(\delta^2 - 1))]^2 \oplus A,$$

where N^l denotes l direct sums of the A-module N.

Let us explicitly describe the previous isomorphism. In order to do that, let us first compute a generalized inverse of the matrix $ext1[2]$ over A:

```
> U:=GeneralizedInverse(ext1[2],A);
```

$$U := \begin{bmatrix} -1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{bmatrix}$$

Let us now introduce the matrix $V = I_4 - U\, ext1[2]$:

```
>   V:=simplify(evalm(1-Mult(U,ext1[2],A)));
```

$$V := \begin{bmatrix} 0 & 0 & 0 & \delta^2+1 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Using the fact that $ext1[2]\, V = 0$, we obtain that the A-morphism $\sigma : E/t(E) \longrightarrow E$ defined by $\sigma(\pi'(\lambda)) = \pi(\lambda\, V)$, where $\pi : A^{1\times 4} \longrightarrow E$ (resp., $\pi' : A^{1\times 4} \longrightarrow E/t(E)$) denotes the canonical projection onto E (resp., $E/t(E)$) and $\lambda \in A^{1\times 4}$, satisfies $\rho \circ \sigma = \mathrm{id}_{E/t(E)}$. For more details, see Theorem 4 of A. Quadrat, D. Robertz, "Parametrizing all solutions of uncontrollable multidimensional linear systems", *Proceedings of* 16^{th} *IFAC World Congress*, Prague (Czech Republic), 04-08/07/05. If we denote by $\{g_i = \rho(f_i)\}_{i=1,\ldots,4}$ a set of generators of the A-module $E/t(E)$, then the A-morphism $\sigma : E/t(E) \longrightarrow E$ is defined by:

$$\begin{cases} \sigma(g_1) = (\delta^2+1)\, f_4, \\ \sigma(g_2) = f_4, \\ \sigma(g_3) = 0, \\ \sigma(g_4) = f_4. \end{cases}$$

Using the relations $Endo[2]\, F = 0$ between the generators f_i's of the A-module E, we obtain that the A-morphism $\chi : \mathrm{id}_E - \sigma \circ \rho : E \longrightarrow E$ is defined by:

$$\begin{cases} \chi(f_1) = f_1 - (\delta^2+1)\, f_4 = -t_1 = t_2, \\ \chi(f_2) = f_2 - f_4 = t_2, \\ \chi(f_3) = f_3 = t_3, \\ \chi(f_4) = f_4 - f_4 = 0. \end{cases}$$

Hence, if we define the A-morphism $\kappa : E \longrightarrow t(E)$ by

$$\begin{cases} \kappa(f_1) = t_2, \\ \kappa(f_2) = t_2, \\ \kappa(f_3) = t_3, \\ \kappa(f_4) = 0, \end{cases}$$

then we get that $\mathrm{id}_E = \sigma \circ \rho + \iota \circ \kappa$. Therefore, we obtain

$$\begin{cases} f_1 = t_2 + (\delta^2+1)\, f_4, \\ f_2 = t_2 + f_4, \\ f_3 = t_3, \\ f_4 = f_4, \end{cases} \qquad (1)$$

which shows that the generators f_i's of E can be expressed in terms of the elements $t_2 = f_2 - f_4 = -t_1$, $t_3 = f_3$ and f_4, a fact proving that $\{t_2, t_3, f_4\}$ is also a family of generators of the A-module E. Using the multiplication table $Endo[3]$ and (1), we can easily obtain the following multiplication table for the new family of generators $\{t_2, t_3, f_4\}$ of E:

$$\begin{cases} t_2 \circ t_2 = 2\,t_2, \\ t_2 \circ t_3 = -2\,d\,t_2, \\ t_2 \circ f_4 = -t_2, \\ t_3 \circ t_2 = 2\,t_3, \\ t_3 \circ t_3 = -2\,d\,t_3, \\ t_3 \circ f_4 = -t_3, \\ f_4 \circ t_2 = -t_2, \\ f_4 \circ t_3 = 2\,d\,t_2 + t_3, \\ f_4 \circ f_4 = t_2 + f_4. \end{cases}$$

We have previously shown that $M \cong M_1 \oplus M_2$. Hence, we have:

$$E = \mathrm{end}_A(M) \cong \mathrm{end}_A(M_1) \oplus \hom_A(M_1, M_2) \oplus \hom_A(M_2, M_1) \oplus \mathrm{end}_A(M_2).$$

Using the fact that $M_1 = A/(A\,(\delta^2 - 1))$, we have $\mathrm{end}_A(M_1) \cong A/(A\,(\delta^2 - 1))$. The fact that M_1 is a torsion A-module and M_2 is torsion-free implies that $\hom_A(M_1, M_2) = 0$. We now need to study $\hom_A(M_2, M_1)$ and $\mathrm{end}_A(M_2)$. Let us denote by $S = (\delta^2 - 1)$ and $T = (\delta^2 + 1 \quad -4\,d\,\delta)$:

```
> S:=submatrix(R_final,1..1,1..1);
```

$$S := \begin{bmatrix} \delta^2 - 1 \end{bmatrix}$$

```
> T:=submatrix(R_final,2..2,2..3);
```

$$T := \begin{bmatrix} \delta^2 + 1 & -4\,d\,\delta \end{bmatrix}$$

Then, $\hom_A(M_2, M_1)$ is defined by:

```
> Morph:=MorphismsConstCoeff(T,S,A);
```

$$Morph := [[\begin{bmatrix} 2\,d \\ \delta \end{bmatrix}], \begin{bmatrix} \delta^2 - 1 \end{bmatrix}]$$

In particular, $\hom_A(M_2, M_1)$ is defined by only one generator h which satisfies $(\delta^2 - 1)\,h = 0$, i.e., $\hom_A(M_2, M_1) \cong A/(A\,(\delta^2 - 1))$.

Finally, let us compute $\mathrm{end}_A(M_2)$:

```
> Endo_T:=MorphismsConstCoeff(T,T,A):
> Endo_T[1];
```

$$[\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 4\,d\,\delta \\ 0 & \delta^2 + 1 \end{bmatrix}]$$

We obtain that the A-module $\mathrm{end}_A(M_2)$ is defined by two generators k_1 and k_2 which satisfy the following A-linear relation:

```
> Endo_T[2];
```

$$\begin{bmatrix} \delta^2 + 1 & -1 \end{bmatrix}$$

As we have the following relation $k_2 = (\delta^2 + 1)\,k_1$, the A-module $\mathrm{end}_A(M_2)$ is generated by k_1 which does not satisfy any other relation. Hence, we get $\mathrm{end}_A(M_2) \cong A$. Hence, we finally find again that $E \cong [A/(A\,(\delta^2 - 1))]^2 \oplus A$.